
    Overview

PhoenixTestPlaywright
Execute PhoenixTest cases in an actual browser via Playwright.
defmodule Features.RegisterTest do
  use PhoenixTest.Playwright.Case,
    async: true,                         # async with Ecto sandbox
    parameterize: [                      # run in multiple browsers in parallel
      %{browser_pool: :chromium},
      %{browser_pool: :firefox}
    ]

  @tag trace: :open                      # replay in interactive viewer
  test "register", %{conn: conn} do
    conn
    |> visit(~p"/")
    |> click_link("Register")
    |> fill_in("Email", with: "f@ftes.de")
    |> click_button("Create an account")
    |> assert_has(".error", text: "required")
    |> screenshot("error.png", full_page: true)
  end
end
Please get in touch with feedback of any shape and size.
Enjoy! Freddy.
P.S. Looking for a standalone Playwright client? See PlaywrightEx.
Getting started
	Add dependency
 # mix.exs
 {:phoenix_test_playwright, "~> 0.14", only: :test, runtime: false}

	Install playwright and browser
 npm --prefix assets i -D playwright
 npx --prefix assets playwright install chromium --with-deps


	Config
 # config/test.exs
 config :phoenix_test, otp_app: :your_app
 config :your_app, YourAppWeb.Endpoint, server: true

	Runtime config
 # test/test_helper.exs
 {:ok, _} = PhoenixTest.Playwright.Supervisor.start_link()
 Application.put_env(:phoenix_test, :base_url, YourAppWeb.Endpoint.url())

	Use in test
 defmodule MyTest do
   use PhoenixTest.Playwright.Case

   # `conn` isn't a `Plug.Conn` but a Playwright session.
   # We use the name `conn` anyway so you can easily switch `PhoenixTest` drivers.
   test "in browser", %{conn: conn} do
     conn
     |> visit(~p"/")
     |> evaluate("console.log('Hey')")
   end
 end

	(Optional) Enable concurrent browser tests with async: true: see Ecto Sandbox

	(Optional) LLM usage rules for AI coding agents (via usage_rules)
 # mix.exs
 def project do
   [
     ...
     usage_rules: usage_rules()
   ]
 end

 defp usage_rules do
   [
     # Option A: inline into a rules file
     file: "AGENTS.md", # or "CLAUDE.md"
     usage_rules: [~r/^phoenix/],
     # Option B: generate as skill (can be used instead of or in addition to file)
     skills: [
       location: ".agents/skills", # or ".claude/skills"
       deps: [:phoenix_test_playwright]
     ]
   ]
 end
 Then run mix usage_rules.sync.


Reference project
github.com/ftes/phoenix_test_playwright_example
The last commit adds a feature test for the phx gen.auth registration page
and runs it in CI (Github Actions).

Configuration
# config/test.exs
config :phoenix_test,
  otp_app: :your_app,
  playwright: [
    browser_pool: :chromium_pool,
    browser_pools: [
      [id: :chromium_pool, browser: :chromium],
      [id: :firefox_pool, browser: :firefox]
    ],
    js_logger: false,
    browser_launch_timeout: 10_000
  ]
See PhoenixTest.Playwright.Config for more details.
You can override some options in your test:
defmodule DebuggingFeatureTest do
  use PhoenixTest.Playwright.Case,
    async: true,
    # Launch new browser for this test suite with custom options below
    browser_pool: false,
    # Show browser and pause 1 second between every interaction
    headless: false,
    slow_mo: :timer.seconds(1)
end
Remote Playwright Server
Connect to a remote Playwright server via WebSocket instead of spawning a local
Node.js driver. Useful for Alpine Linux containers (glibc issues) or containerized CI.
# mix.exs
{:websockex, "~> 0.4", only: :test}

# config/test.exs
config :phoenix_test, playwright: [ws_endpoint: "ws://localhost:3000", browser_pool: false]

# or, to enable via environment variable
config :phoenix_test, playwright: [ws_endpoint: System.get_env("PLAYWRIGHT_WS_ENDPOINT"), browser_pool: false]
# Start Playwright server
docker run -p 3000:3000 --rm --init -it --workdir /home/pwuser --user pwuser mcr.microsoft.com/playwright:v1.58.0-noble /bin/sh -c "npx -y playwright@1.58.0 run-server --port 3000 --host 0.0.0.0"

The browser type is automatically appended as a query parameter (e.g., ?browser=chromium).
Traces
Playwright traces record a full browser history, including 'user' interaction, browser console, network transfers etc.
Traces can be explored in an interactive viewer for debugging purposes.
Manually
@tag trace: :open
test "record a trace and open it automatically in the viewer" do
Automatically for failed tests in CI
# config/test.exs
config :phoenix_test, playwright: [trace: System.get_env("PW_TRACE", "false") in ~w(t true)]
# .github/workflows/elixir.yml
run: "mix test || if [[ $? = 2 ]]; then PW_TRACE=true mix test --failed; else false; fi"
Step annotations
Playwright traces support grouping labelled test steps and
assigning them source code locations. This makes it easier to see what a test
is doing and where. These groups are visible in the Playwright trace viewer.
test "user registration", %{conn: conn} do
  conn
  |> visit(~p"/")
  |> step("Submit registration form", fn conn ->
    conn
    |> fill_in("Email", with: "user@example.com")
    |> fill_in("Password", with: "secret")
    |> click_button("Sign up")
  end)
  |> assert_has(".flash", text: "Welcome!")
end
Screenshots
Manually
|> visit(~p"/")
|> screenshot("home.png")    # captures entire page by default, not just viewport
Automatically for failed tests in CI
# config/test.exs
config :phoenix_test, playwright: [screenshot: System.get_env("PW_SCREENSHOT", "false") in ~w(t true)]
# .github/workflows/elixir.yml
run: "mix test || if [[ $? = 2 ]]; then PW_SCREENSHOT=true mix test --failed; else false; fi"
Logging in
For username/password login, just visit the login page and fill in the credentials:
conn
|> visit(~p"/users/log_in")
|> fill_in("Email", with: "user@example.com")
|> fill_in("Password", with: "password123")
|> click_button("Sign in")
For magic link / passwordless login, see the Emails section below.
Emails
If you want to verify the HTML of sent emails in your feature tests,
consider using Plug.Swoosh.MailboxPreview.
The iframe used to render the email HTML body makes this slightly tricky:
|> visit(~p"/dev/mailbox")
|> click_link("Confirmation instructions")
|> within("iframe >> internal:control=enter-frame", fn conn ->
  conn
  |> click_link("Confirm account")
  |> click_button("Confirm my account")
  |> assert_has("#flash-info", text: "User confirmed")
end)
Common problems
Test failure in CI (timeout)
	Limit concurrency: config :phoenix_test, playwright: [browser_pools: [[size: 1]]] or mix test --max-cases 1 for GitHub CI shared runners
	Increase timeout: config :phoenix_test, playwright: [timeout: :timer.seconds(4)]
	More compute power: e.g. x64 8-core GitHub runner

LiveView not connected
|> visit(~p"/")
|> assert_has("body .phx-connected")
# now continue, playwright has waited for LiveView to connect
LiveComponent not connected
<div id="my-component" data-connected={connected?(@socket)}>
|> visit(~p"/")
|> assert_has("#my-component[data-connected]")
# now continue, playwright has waited for LiveComponent to connect
Browser version mismatch
If you've installed a browser but can't run tests
(Executable doesn't exist at .../ms-playwright/chromium_headless_shell-1208/),
you probably used the wrong playwright JS version to install the browser.
Each playwright JS version pins a specific browser version.
Tests are run using ./assets/node_modules/playwright
(see assets_dir in PhoenixTest.Playwright.Config).
Make sure to use that same playwright JS version to install the browser,
e.g. via npx --prefix assets playwright install.
Ecto Sandbox
Make sure you have followed the advanced set up instructions for Phoenix.Ecto.SQL.Sandbox
	with LiveViews
	with Channels
	with Ash authentication: use on_mount_prepend

PhoenixTest.Playwright.Case takes care of the rest. It starts the
sandbox under a separate process than your test and uses
ExUnit.Callbacks.on_exit/1 to ensure the sandbox is shut down afterward. It
also sends a User-Agent header with the
Phoenix.Ecto.SQL.Sandbox metadata for your Ecto repos. This allows
the sandbox to be shared with the LiveView and other processes which need to
use the database inside the same transaction as the test. It also allows for
concurrent browser tests.
defmodule MyTest do
  use PhoenixTest.Playwright.Case, async: true
end
Ownership errors with LiveViews
Unlike Phoenix.LiveViewTest, which controls the lifecycle of LiveView
processes being tested, Playwright tests may end while such processes are
still using the sandbox.
In that case, you may encounter ownership errors like:
** (DBConnection.OwnershipError) cannot find owner for ...
To prevent this, the ecto_sandbox_stop_owner_delay option allows you to delay the
sandbox owner's shutdown, giving LiveViews and other processes time to close
their database connections. The delay happens during
ExUnit.Callbacks.on_exit/1, which blocks the running of the next test, so
it affects test runtime as if it were a Process.sleep/1 at the end of your
test.
So you probably want to use as small a delay as you can, and only for the
tests that need it, using @tag (or @describetag or @moduletag) like:
@tag ecto_sandbox_stop_owner_delay: 100 # 100ms
test "does something" do
  # ...
end
If you want to set a global default, you can:
# config/test.exs
config :phoenix_test, playwright: [
  ecto_sandbox_stop_owner_delay: 50  # 50ms
]
Missing Playwright features
This library adds functions beyond the standard PhoenixTest API (e.g. screenshot/3, evaluate/2, click_link/4),
but it does not wrap the entire Playwright API.
You can add any missing functionality yourself using unwrap/2 with
PlaywrightEx modules (Frame, Selector, Page, BrowserContext),
and the Playwright source.
If you think others might benefit, please open a PR.
Here is some inspiration:
def choose_styled_radio_with_hidden_input_button(conn, label, opts \\ []) do
  opts = Keyword.validate!(opts, exact: true)
  PhoenixTest.Playwright.click(conn, PlaywrightEx.Selector.text(label, opts))
end

def assert_a11y(conn) do
  PlaywrightEx.Frame.evaluate(conn.frame_id, expression: A11yAudit.JS.axe_core(), timeout: timeout())
  {:ok, json} = PlaywrightEx.Frame.evaluate(conn.frame_id, expression: "axe.run()", timeout: timeout())
  results = A11yAudit.Results.from_json(json)
  A11yAudit.Assertions.assert_no_violations(results)

  conn
end

def within_iframe(conn, selector \\ "iframe", fun) when is_function(fun, 1) do
  within(conn, "#{selector} >> internal:control=enter-frame", fun)
end
Contributing
To run the tests locally, you'll need to:
	Check out the repo
	Run mix setup. This will take care of setting up your dependencies, installing the JavaScript dependencies (including Playwright), and compiling the assets.
	Run mix test or, for a more thorough check that matches what we test in CI, run mix check
	Run mix test.websocket to run all tests against a 'remote' playwright server via websocket. Docker needs to be installed. A container is started via testcontainers.

Conventions
	Follows PhoenixTest API. Only add new public functions when strictly necessary for browser-specific interaction (e.g., screenshots, JS evaluation).
	Do not edit upstream tests. Files under test/phoenix_test/upstream/ are mirrored from phoenix_test and must not be modified. Playwright-specific tests go in test/phoenix_test/playwright_test.exs or other files outside upstream/.

Playwright internals
Playwright's implementation is split between a client (Node.js API) and a server (browser protocol layer). The Playwright docs describe the public API but don't reflect this split. When reading Playwright source code, it can help to look at the TypeScript sources directly: client and server (locally under priv/static/assets/node_modules/playwright-core/lib/).


  

    Changelog

All notable changes to this project will be documented in this file.
The format is based on Keep a Changelog.
[0.15.0] 2026-06-19
Added
	assert_screenshot/3: visual regression testing. Commit 18353cc, @Wigny, #181

Changed
	Validate keyword options instead of silently ignoring unknown keys. Commit aacc30c, #180
	visit/3: Accept all supported options for goto. Commit 7f126e0, @s3cur3, #171
	assert_has/3, refute_has/3: Honour options checked, selected. Commit 43c689e, #179

Fixed
	submit/1: submit form properly if last interacted input was a 'select' (don't submit via 'Enter' press due to varying OS behaviour). Commit [d85dfc8], #165

[0.14.0] 2026-05-05
Added
	Support phoenix_test v0.11.1. Commit f60d668	Add assert_download/2. Commit 244a0f5


	PhoenixTest.Playwright.evaluate/3-4: support function expressions via is_function: true and pass arguments via arg. Commit 2730ab5
	Support multiple sessions in async: false suites. Commit 270d1c2, @rubysolo

[0.13.0] 2026-03-06
Added
	Support phoenix_test v0.10.0	Add reload_page/2



[0.12.1] 2026-02-23
Added
	Recommend browser_pool: false when using ws_endpoint (remote server has a single pre-launched browser). Commit 60f4b6f

Changed
	Use browser_pool: false (instead of nil) to disable browser pooling. Commit 74eb3a7

Fixed
	Case.new_session/2: 'already allowed' error (regression in v0.12.0, commit 0a8538c). Commit 281d71a

[0.12.0] 2026-02-22
Added
	PhoenixTest.Playwright.evaluate/2-4: evaluate JavaScript snippets in the browser. Commit eead742
	usage-rules.md for LLM coding agents (via usage_rules). Commit 80eb40c
	timeout option for custom Playwright public functions (evaluate, step, click, etc.). Commit 0799600
	drag/3: pass additional options through to Playwright. Commit fc3f15c

Fixed
	assert_path/refute_path now work after LiveView patches and navigations (push_patch, push_navigate, etc.). Commit 2fdd358
	Restore helpful error message when browser_launch_timeout is too small (lost during browser pool introduction). Commit 596f938

[0.11.1] 2026-02-10
Fixed
	Fix regression: Config validation error if browser_pools not set. Commit 987828e
	Fix regression: Don't rely on package-lock.json for bun. Commit 8b342a8

[0.11.0] 2026-02-09
Added
	Support remote Playwright server via WebSocket. Commit 396fbdc, @carsoncall

Changed
	Browser pools: Fall back to global options (headless, slow_mo etc). Commit 72b9799

[0.10.1] 2026-01-30
Added
	PhoenixTest.Playwright.step/3: label groups of actions in a trace with automatic source file and line. Commit 3eaeb5a, @nathanl

[0.10.0] 2025-12-24
Fixed
	Correctly handle custom options via browser_context_opts and browser_page_opts config. Commit 2b80c87, @melucasleite

Added
	Support timeout opt in PhoenixTest.Playwright.visit/3. Commit 7073add, @s3cur3

Removed
	runner config: Don't fequired npm or bun, invoke cli.js directly. Commit 1605dce 

[0.10.0-rc.0] 2025-11-19
Breaking changes
	Use browser pool by default, instead of starting new browser per test suite. Commit 095e216  # test_helper.exs
+ {:ok, _} = PhoenixTest.Playwright.Supervisor.start_link()
  Application.put_env(:phoenix_test, :base_url, MyAppWeb.Endpoint.url())

	Changes required if internals (beyond PhoenixTest and PhoenixTest.Playwright modules) were used. Phoenix-agnostic modules moved to PlaywrightEx, with slight API changes. Example:- |> unwrap(& {:ok, _} = PhoenixTest.Playwright.Frame.click(&1.frame_id, selector))
+ |> unwrap(& {:ok, _} = PlaywrightEx.Frame.click(&1.frame_id, selector: selector, timeout: timeout())


Changed
	Ecto sandbox ownership: Use a separate sandbox owner process instead of the test process. This reduces ownership errors when LiveViews continue to use database connections after the test terminates. Commit 3b54699

Added
	Config option ecto_sandbox_stop_owner_delay: Delay in milliseconds before shutting down the Ecto sandbox owner. Use when LiveViews or other processes need time to stop using the connections. Commit 2f4a8cf

[0.9.1] 2025-10-29
Added
	Browser pooling (opt-in): Reduced memory, higher speed. Commit 00e75c6

[0.9.0] 2025-10-26
Fixed
	assert_has/refute_has: don't raise if multiple nodes found when using value option (playwright strict mode). Commit 73ebf10

Changed
	Return result tuples from all playwright channel functions for consistency and to surface errors early. Commit ae63989	Most notably may affect callers of Frame.evaluate/3



Added
	Register custom selector engines via new config option selector_engines. Commit 73ebf10
	Import click/2 in Playwright.Case. Commit 968d5cd
	Add drag and drop: drag(source, to: target). Commit f4161bd

[0.8.0] 2025-09-17
Removed
	Config option cli. Use assets_dir instead. Commit 9e95e54, @Wigny

Added
	Support bunx JS runner. Added config options runner and assets_dir. Commit 9e95e54, @Wigny
	Support missing assert_has/refute_has options: label and value. Commit 2e92cbe
	Support phoenix_test 0.8 (lazy_html). Commit 1074cde

Changed
	Include source location when logging javascript errors and browser console logs. Commit 6b148f, @tomfarm
	Only consider visible inputs for fill_in etc. Commit 86c2e3d
	Speed up refute_has: Use playwright browser internal retry. Commit aac0497

[0.7.1] 2025-07-01
Added
	Config option executable_path: allow using existing browser executable instead of bundled browser (e.g. on NixOS). Commit 15df46

Fixed
	refute_has/3 add retry, don't fail if element initially found. Commit 7bd49b

[0.7.0] 2025-06-16
Added
	Dialog handling. Commit 4eadea	Config option accept_dialogs (default: true)
	PhoenixTest.Playwright.with_dialog/3 for conditional handling



Removed
	Connection.received/1. Commit 4eadea	Use EventRecorder instead



[0.6.3] 2025-05-05
Added
	Add locale to avoid console warnings. Commit [becf5e6], @peaceful-james

Fixed
	Trigger phx-change event for input with phx-debounce="blur". Commit 72edd9

[0.6.2] 2025-04-24
Changed
	Frame.evaluate/3: Don't transform map keys in return value. Example: js {camelCase: {a: 1}} -> ex %{"camelCase": %{"a": 1}}. Previously attempted to underscore and atom-ize keys, which led to issue #24. Commit 5ff530

[0.6.1] 2025-04-02
Added
	Support relevant phoenix_test 0.6 features	✅ Deprecate select with :from in favor of :option (handled by PhoenixTest)
	✅ Allow nesting of within/3
	✅ Allow calling visit/2 sequentially (was already supported)
	❌ Raise error when visiting a defined route: In a feature test, we assert on the rendered page, not the status code.



[0.6.0] 2025-03-18
Added
	Add and remove cookies: add_cookies/2, add_session_cookie/3, clear_cookies/{1,2} @peaceful-james
	Add option browser_launch_timeout for more fine-grained control (is typically a lot slower than other actions) @s3cur3

[0.5.0] 2025-02-14
Added
	Docs: Document and validate global and per-test configuration
	Docs: Document non-standard driver functions (click/4, click_button/4 etc.). Also, exclude standard driver functions from docs.
	Config: Override config via Case opts, e.g. use PhoenixTest.Playwright.Case, headless: false
	Keyboard simulation: type/{3,4} and press/{3,4}

Changed
	Renamed: PheonixTest.Case to PhoenixTest.Playwright.Case   defmodule MyTest do
-    use PhoenixTest.Case, async: true
+    use PhoenixTest.Playwright.Case, async: true
-    @moduletag :playwright


[0.4.0] 2025-02-03
Added
	Screenshots: screenshot/{2,3} function and screenshot: true config for auto-capture @s3cur3

Changed
	Config: flattened list (remove nested browser config), override via top-level ExUnit @tag ...s (remove nested @tag playwright: [...])   # config/test.exs
   config :phoenix_test,
     playwright: [
-      browser: [browser: :chromium, headless: false, slow_mo: 0]
+      browser: :chromium,
+      headless: false,
+      slow_mo: 0


[0.3.0] 2025-01-26
Changed
	Auto-convert case of playwright messages keys (snake_case to camelCase)

[0.2.1] 2025-01-17
Added
	Add more config options (browser, JS console) @s3cur3

Changed
	Improve error messages @s3cur3
	Improve setup and docs for contributors @s3cur3

[0.2.0] 2025-01-09
Added
	support phoenix_test@0.5, elixir@1.18, phoenix_live_view@1.0

[0.1.5] 2024-12-15
Added
	@tag trace: :open to auto open recorded Playwright trace in viewer



  

    
PhoenixTest.Playwright 
    



      
Playwright driver for PhoenixTest.
This module implements the PhoenixTest driver protocol, running tests in a real browser
via Playwright. The conn in tests is not a Plug.Conn but a %PhoenixTest.Playwright{}
struct holding the Playwright session state (page, frame, browser context, etc.).
It also provides browser-specific functions beyond the standard PhoenixTest API,
such as screenshot/3, evaluate/2, type/3, press/3, and drag/3.
See the README for getting started, configuration, and troubleshooting.
Missing Playwright features
See the README.

        add_cookies(conn, cookies)

      


        Add cookies to the browser context, using Plug.Conn.put_resp_cookie/3



    


    
      
        add_session_cookie(conn, cookie, session_options)

      


        Add a Plug.Session cookie to the browser context.



    


    
      
        assert_download(conn, filename_or_fun)

      


    


    
      
        assert_screenshot(conn, name, opts \\ [])

      


        Visual regression assertion: compares a screenshot of the current page against a stored baseline.



    


    
      
        clear_cookies(conn, opts \\ [])

      


        Removes all cookies from the context.



    


    
      
        click(conn, selector)

      


        See click/4.



    


    
      
        click(conn, selector, text, opts \\ [])

      


        Click an element that is not a link or button.
Otherwise, use click_link/4 and click_button/4.



    


    
      
        click_button(conn, selector \\ nil, text, opts \\ [])

      


        Like PhoenixTest.click_button/3, but allows exact text match.



    


    
      
        click_link(conn, selector \\ nil, text, opts \\ [])

      


        Like PhoenixTest.click_link/3, but allows exact text match.



    


    
      
        drag(conn, source_selector, opts)

      


        Drag and drop a source element to a target element.



    


    
      
        evaluate(conn, expression)

      


        Evaluates a JavaScript expression in the page context.



    


    
      
        evaluate(conn, expression, fun)

      


    


    
      
        evaluate(conn, expression, opts, fun)

      


    


    
      
        press(conn, selector, key, opts \\ [])

      


        Focuses the matching element and presses a combination of the keyboard keys.



    


    
      
        screenshot(conn, file_path, opts \\ [])

      


        Takes a screenshot of the current page and saves it to the given file path.



    


    
      
        step(conn, title, fun)

      


        Label a step in the Playwright trace.



    


    
      
        type(conn, selector, text, opts \\ [])

      


        Focuses the matching element and simulates user typing.



    


    
      
        unwrap(conn, fun)

      


        See PhoenixTest.unwrap/2.



    


    
      
        visit(conn, path, opts)

      


        Like PhoenixTest.visit/2, but with a custom timeout.



    


    
      
        with_dialog(session, callback, fun)

      


        Handle browser dialogs (alert(), confirm(), prompt()) while executing the inner function.

          @type css_selector() :: String.t()

          @type playwright_selector() :: String.t()

          @type selector() :: playwright_selector() | css_selector()

Add cookies to the browser context, using Plug.Conn.put_resp_cookie/3
Note that for signed cookies the signing salt is not configurable.
As such, this function is not appropriate for signed Plug.Session cookies.
For signed session cookies, use add_session_cookie/3
  A cookie's value must be a binary unless the cookie is signed/encrypted
Cookie fields
	key	type	description
	:name	binary()	
	:value	binary()	
	:url	binary()	(optional) either url or domain / path are required
	:domain	binary()	(optional) either url or domain / path are required
	:path	binary()	(optional) either url or domain / path are required
	:max_age	float()	(optional) The cookie max age, in seconds.
	:http_only	boolean()	(optional)
	:secure	boolean()	(optional)
	:encrypt	boolean()	(optional)
	:sign	boolean()	(optional)
	:same_site	binary()	(optional) one of "Strict", "Lax", "None"

Two of the cookie fields mean nothing to Playwright. These are:
	:encrypt
	:sign

The :max_age cookie field means the same thing as documented in Plug.Conn.put_resp_cookie/4.
The :max_age value is used to infer the correct expires value that Playwright requires.
See https://playwright.dev/docs/api/class-browsercontext#browser-context-add-cookies

Add a Plug.Session cookie to the browser context.
This is useful for emulating a logged-in user.
Note that that the cookie :value must be a map, since we are using
Plug.Conn.put_session/3 to write each of value's key-value pairs
to the cookie.
The session_options are exactly the same as the opts used when
writing plug Plug.Session in your router/endpoint module.
Examples
|> add_session_cookie(
  [value: %{user_token: Accounts.generate_user_session_token(user)}],
  MyAppWeb.Endpoint.session_options()
)

  



  
    
      
    
    
      assert_download(conn, filename_or_fun)



        
          
        

    

  


  


  



    

  
    
      
    
    
      assert_screenshot(conn, name, opts \\ [])



        
          
        

    

  


  

      

          @spec assert_screenshot(t(), String.t(),
  full_page: boolean(),
  omit_background: boolean(),
  animations: term(),
  caret: term(),
  clip: map(),
  selector: binary(),
  max_diff_pixels: non_neg_integer(),
  max_diff_pixel_ratio: float(),
  threshold: float(),
  scale: term(),
  mask: [binary()],
  mask_color: binary(),
  snapshot_dir: binary(),
  timeout: non_neg_integer()
) :: t()


      


Visual regression assertion: compares a screenshot of the current page against a stored baseline.
name is a relative path within :snapshot_dir (see PhoenixTest.Playwright.Config), including
the file extension — for example "home.png" or "auth/login.png". Subdirectories are created
automatically.
On first run (no baseline exists), the screenshot is saved as the new baseline and the test passes.
On subsequent runs, the live screenshot is compared pixel-by-pixel against the baseline.
On mismatch, a diff image is saved to <snapshot_dir>/__diff__/<name> and the test fails.
To update a baseline, delete the snapshot file and re-run the test.
Options
	:full_page (boolean/0) - When true, takes a screenshot of the full scrollable page, instead of the currently visible viewport. Defaults to false.

	:omit_background (boolean/0) - Hides default white background and allows capturing screenshots with transparency. Not applicable to jpeg images. Defaults to false.

	:animations - When set to "disabled", stops CSS animations, CSS transitions and Web Animations. Animations get different treatment depending on their duration: finite animations are fast-forwarded to completion, so they'll fire transitionend event. Infinite animations are canceled to initial state, and then played over after the screenshot. Defaults to "disabled".

	:caret - When set to "hide", screenshot will hide text caret. When set to "initial", text caret behavior will not be changed. Defaults to "hide".

	:clip (map/0) - An object which specifies clipping of the resulting image. %{x: float, y: float, width: float, height: float}.

	:selector (String.t/0) - Scope the screenshot to a CSS or Playwright selector instead of the full page.

	:max_diff_pixels (non_neg_integer/0) - An acceptable amount of pixels that could be different. Unset by default.

	:max_diff_pixel_ratio (float/0) - An acceptable ratio of pixels that are different to the total amount of pixels, between 0 and 1. Unset by default.

	:threshold (float/0) - An acceptable perceived color difference in the YIQ color space between the same pixel in compared images, between zero (strict) and one (lax). Defaults to 0.2.

	:scale - When set to "css", screenshot will have a single pixel per each css pixel on the page. For high-dpi devices, this will keep screenshots small. Using "device" option will produce a single pixel per each device pixel. Defaults to "css".

	:mask (list of String.t/0) - Specify selectors that should be masked when the screenshot is taken. Masked elements will be overlaid with a pink box #FF00FF (customized by mask_color) that completely covers its bounding box.

	:mask_color (String.t/0) - Specify the color of the overlay box for masked elements, in CSS color format. Default color is pink #FF00FF.

	:snapshot_dir (String.t/0) - Override the global :snapshot_dir config for this call.

	:timeout (non_neg_integer/0) - Maximum wait time in milliseconds. Defaults to the configured timeout.


Examples
|> assert_screenshot("home.png")
|> assert_screenshot("auth/login.png", full_page: false)
|> assert_screenshot("hero.png", selector: ".hero-section")

  



    

  
    
      
    
    
      clear_cookies(conn, opts \\ [])



        
          
        

    

  


  

      

          @spec clear_cookies(t(), [{:timeout, non_neg_integer()}]) :: t()


      


Removes all cookies from the context.
Options
	:timeout (non_neg_integer/0) - Maximum wait time in milliseconds. Defaults to the configured timeout.


  



  
    
      
    
    
      click(conn, selector)



        
          
        

    

  


  

      

          @spec click(t(), selector()) :: t()


      


See click/4.

  



    

  
    
      
    
    
      click(conn, selector, text, opts \\ [])



        
          
        

    

  


  

      

          @spec click(t(), selector(), String.t(), exact: boolean(), timeout: non_neg_integer()) ::
  t()


      


Click an element that is not a link or button.
Otherwise, use click_link/4 and click_button/4.
Options
	:exact (boolean/0) - Exact or substring text match. The default value is false.

	:timeout (non_neg_integer/0) - Maximum wait time in milliseconds. Defaults to the configured timeout.


Examples
|> click(Selector.menuitem("Edit"))
|> click("summary", "(expand)", exact: false)

  



    

    

  
    
      
    
    
      click_button(conn, selector \\ nil, text, opts \\ [])



        
          
        

    

  


  

Like PhoenixTest.click_button/3, but allows exact text match.
Options
	:exact (boolean/0) - Exact or substring text match. The default value is false.

	:timeout (non_neg_integer/0) - Maximum wait time in milliseconds. Defaults to the configured timeout.



  



    

    

  
    
      
    
    
      click_link(conn, selector \\ nil, text, opts \\ [])



        
          
        

    

  


  

Like PhoenixTest.click_link/3, but allows exact text match.
Options
	:exact (boolean/0) - Exact or substring text match. The default value is false.

	:timeout (non_neg_integer/0) - Maximum wait time in milliseconds. Defaults to the configured timeout.



  



  
    
      
    
    
      drag(conn, source_selector, opts)



        
          
        

    

  


  

      

          @spec drag(t(), selector(),
  to: selector(),
  playwright: keyword(),
  timeout: non_neg_integer()
) :: t()


      


Drag and drop a source element to a target element.
Options
	:to (selector/0) - Required. The target selector.

	:playwright (keyword/0) - Additional options passed to frame.dragAndDrop. The default value is [].

	:timeout (non_neg_integer/0) - Maximum wait time in milliseconds. Defaults to the configured timeout.


Examples
|> drag("#source", to: "#target")
|> drag(Selector.text("Draggable"), to: Selector.text("Target"))

  



  
    
      
    
    
      evaluate(conn, expression)



        
          
        

    

  


  

      

          @spec evaluate(t(), String.t()) :: t()


      


Evaluates a JavaScript expression in the page context.
When a callback function is given, it receives the JavaScript result. This
is useful for assertions or side effects without breaking the pipe chain.
Options
	:timeout (non_neg_integer/0) - Maximum wait time in milliseconds. Defaults to the configured timeout.

	:is_function (boolean/0) - Whether the expression is a function. The default value is false.

	:arg (term/0) - Optional argument to pass to the function. The default value is nil.


Examples
conn
|> evaluate("window.scrollTo(0, document.body.scrollHeight)")

conn
|> evaluate("selectors => selectors.forEach(s => document.querySelector(s).remove())", is_function: true, arg: ["h1"])

conn
|> evaluate("document.title", & assert &1 =~ "Dashboard")

  



  
    
      
    
    
      evaluate(conn, expression, fun)



        
          
        

    

  


  

      

          @spec evaluate(
  t(),
  String.t(),
  [timeout: non_neg_integer(), is_function: boolean(), arg: term()]
  | (any() -> any())
) :: t()


      



  



  
    
      
    
    
      evaluate(conn, expression, opts, fun)



        
          
        

    

  


  

      

          @spec evaluate(
  t(),
  String.t(),
  [timeout: non_neg_integer(), is_function: boolean(), arg: term()],
  (any() -> any())
) :: t()


      



  



    

  
    
      
    
    
      press(conn, selector, key, opts \\ [])



        
          
        

    

  


  

      

          @spec press(t(), selector(), String.t(),
  delay: non_neg_integer(),
  timeout: non_neg_integer()
) :: t()


      


Focuses the matching element and presses a combination of the keyboard keys.
Use type/4 if you don't need to press special keys.
Examples of supported keys:
F1 - F12, Digit0- Digit9, KeyA- KeyZ, Backquote, Minus, Equal, Backslash, Backspace, Tab, Delete, Escape, ArrowDown, End, Enter, Home, Insert, PageDown, PageUp, ArrowRight, ArrowUp
Modifiers are also supported:
Shift, Control, Alt, Meta, ShiftLeft, ControlOrMeta
Combinations are also supported:
Control+o, Control++, Control+Shift+T
Options
	:delay (non_neg_integer/0) - Time to wait between keydown and keyup in milliseconds. The default value is 0.

	:timeout (non_neg_integer/0) - Maximum wait time in milliseconds. Defaults to the configured timeout.


Examples
|> press("#id", "Control+Shift+T")
|> press(Selector.button("Submit"), "Enter")

  



    

  
    
      
    
    
      screenshot(conn, file_path, opts \\ [])



        
          
        

    

  


  

      

          @spec screenshot(t(), String.t(),
  full_page: boolean(),
  omit_background: boolean(),
  timeout: non_neg_integer()
) :: t()


      


Takes a screenshot of the current page and saves it to the given file path.
The file type will be inferred from the file extension on the path you provide.
The file is saved in :screenshot_dir, see PhoenixTest.Playwright.Config.
Options
	:full_page (boolean/0) - The default value is true.

	:omit_background (boolean/0) - Only applicable to .png images. The default value is false.

	:timeout (non_neg_integer/0) - Maximum wait time in milliseconds. Defaults to the configured timeout.


Examples
|> screenshot("my-screenshot.png")
|> screenshot("my-test/my-screenshot.jpg")

  



  
    
      
    
    
      step(conn, title, fun)


        (macro)


        
          
        

    

  


  

Label a step in the Playwright trace.
This is useful for marking custom helper functions or complex multi-step operations
so they appear as distinct steps in the trace viewer for easier debugging.
Steps can be nested.
Their source location is noted in the trace.
Examples
def complete_checkout(conn, user_email) do
  conn
  |> sign_in_as(user_email)
  |> step("Check out", fn conn ->
    conn
    |> step("Fill shipping information", fn conn ->
      conn
      |> fill_in("Address", with: "123 Main St")
      |> fill_in("City", with: "Portland")
      |> click_button("Continue")
    end)
    |> step("Fill payment information", fn conn ->
      conn
      |> fill_in("Card number", with: "4242424242424242")
      |> fill_in("CVV", with: "123")
      |> click_button("Place order")
    end)
  end)
end

defp sign_in_as(conn, user_email) do
  conn
  |> step("Sign in as #{user_email}", fn conn ->
    conn
    |> fill_in("Email", with: user_email)
    |> fill_in("Password", with: "password123")
    |> click_button("Sign In")
  end)
end

  



    

  
    
      
    
    
      type(conn, selector, text, opts \\ [])



        
          
        

    

  


  

      

          @spec type(t(), selector(), String.t(),
  delay: non_neg_integer(),
  timeout: non_neg_integer()
) :: t()


      


Focuses the matching element and simulates user typing.
In most cases, you should use PhoenixTest.fill_in/4 instead.
Options
	:delay (non_neg_integer/0) - Time to wait between key presses in milliseconds. The default value is 0.

	:timeout (non_neg_integer/0) - Maximum wait time in milliseconds. Defaults to the configured timeout.


Examples
|> type("#id", "some text")
|> type(Selector.role("heading", "Untitled"), "New title")

  



  
    
      
    
    
      unwrap(conn, fun)



        
          
        

    

  


  

      

          @spec unwrap(t(), (%{context_id: any(), page_id: any(), frame_id: any()} -> any())) ::
  t()


      


See PhoenixTest.unwrap/2.
Invokes fun with various Playwright IDs.
These can be used to interact with the Playwright
BrowserContext,
Page and
Frame.
Examples
|> unwrap(fn %{page_id: page_id} -> PlaywrightEx.subscribe(page_id) end)

  



  
    
      
    
    
      visit(conn, path, opts)



        
          
        

    

  


  

      

          @spec visit(t(), String.t(),
  timeout: non_neg_integer(),
  referer: binary(),
  wait_until: term()
) :: t()


      


Like PhoenixTest.visit/2, but with a custom timeout.
Options
	:timeout (non_neg_integer/0) - Maximum wait time in milliseconds. Defaults to the configured timeout.

	:referer (String.t/0) - Referer header value. If provided, takes preference over the referer header set via the extra_http_headers option to PlaywrightEx.Browser.new_context/2.

	:wait_until - When to consider the operation succeeded. One of:
	"load" (default) — fires when the load event is fired.
	"domcontentloaded" — fires when the DOMContentLoaded event is fired.
	"networkidle" — discouraged — fires when there are no network connections for at least 500 ms. Rely on web assertions instead for testing readiness.
	"commit" — fires when the network response is received and the document has started loading. The default value is "load".




  



  
    
      
    
    
      with_dialog(session, callback, fun)



        
          
        

    

  


  

Handle browser dialogs (alert(), confirm(), prompt()) while executing the inner function.
Note: Add @tag accept_dialogs: false before tests that call this function.
Otherwise, all dialogs are accepted by default.
Callback return values
The callback may return one of these values:
	:accept -> accepts confirmation dialog
	{:accept, prompt_text} -> accepts prompt dialog with text
	:dismiss -> dismisses dialog
	Any other value will ignore the dialog

Examples
@tag accept_dialogs: false
test "conditionally handle dialog", %{conn: conn} do
conn
  |> visit("/")
  |> with_dialog(
    fn
      %{message: "Are you sure?"} -> :accept
      %{message: "Enter the magic number"} -> {:accept, "42"}
      %{message: "Self destruct?"} -> :dismiss
    end,
    fn conn ->
      conn
      |> click_button("Delete")
      |> assert_has(".flash", text: "Deleted")
    end
  end)
end

  


        

      


  

    
PhoenixTest.Playwright.Case 
    



      
ExUnit case module to assist with browser based tests.
PhoenixTest.Playwright and PhoenixTest.Playwright.Config explain
how to use and configure this module.
If the default setup behaviour and order does not suit you, consider
	using config opt browser_context_opts, which are passed to PlaywrightEx.Browser.new_context/2
	using config opt browser_page_opts, which are passed to PlaywrightEx.BrowserContext.new_page/2
	implementing your own Case (the setup functions in this module are public for your convenience)


      


      
        Summary


  
    Functions
  


    
      
        do_setup(context)

      


        Merges the ExUnit context with PhoenixTest.Playwright.Config opts.
Uses the result to create a new browser context and page.
Adds :conn to the context.



    


    
      
        do_setup_all(context)

      


        Merges the ExUnit context with PhoenixTest.Playwright.Config opts.
Uses the result to launch the browser.
Adds :browser_id to the context.



    


    
      
        new_session(config, context)

      


    





      


      
        Functions


        


  
    
      
    
    
      do_setup(context)



        
          
        

    

  


  

Merges the ExUnit context with PhoenixTest.Playwright.Config opts.
Uses the result to create a new browser context and page.
Adds :conn to the context.

  



  
    
      
    
    
      do_setup_all(context)



        
          
        

    

  


  

Merges the ExUnit context with PhoenixTest.Playwright.Config opts.
Uses the result to launch the browser.
Adds :browser_id to the context.

  



  
    
      
    
    
      new_session(config, context)



        
          
        

    

  


  


  


        

      


  

    
PhoenixTest.Playwright.Config 
    



      
Configuration options for the Playwright driver.
Most should be set globally in config/tests.exs.
Some can be overridden per test.
All options:
	:accept_dialogs (boolean/0) - Accept browser dialogs (alert(), confirm(), prompt()). The default value is true.

	:assets_dir (binary/0) - The directory where the JS assets are located and the Playwright CLI is installed.
Playwright version 1.61.0 or newer is recommended.
Alternatively, use ws_endpoint to connect to a remote Playwright server instead, in which case no local node and playwright is required and assets_dir is ignored. The default value is "./assets".

	:browser (:android | :chromium | :electron | :firefox | :webkit) - The default value is :chromium.

	:browser_context_opts - Additional arguments passed to Playwright Browser.newContext.
E.g. [http_credentials: %{username: "a", password: "b"}]. The default value is [].

	:browser_launch_opts (keyword/0) - Additional arguments passed to Playwright browserType.launch.
E.g. [args: ["--use-fake-ui-for-media-stream", "--use-fake-device-for-media-stream"]].
Can't be used with remote browser (ws_endpoint). The default value is [].

	:browser_launch_timeout (non_neg_integer/0) - The default value is 4000.

	:browser_page_opts - Additional arguments passed to Playwright Browser.newPage.
(E.g. [accept_downloads: false]. The default value is [].

	:browser_pool (atom | false) - Reuse a browser from this pool instead of launching a new browser per test suite.
false to disable pooling and launch a new browser per test suite. The default value is :default_pool.

	:browser_pool_checkout_timeout (non_neg_integer/0) - The default value is 60000.

	:browser_pools (list of non-empty keyword/0) - Supported keys:
	:id (atom/0) - Required.

	:size (integer/0) - The default value is System.schedulers_online() / 2.

	:browser (:android | :chromium | :electron | :firefox | :webkit) - The default value is :chromium.

	:browser_launch_opts (keyword/0) - Additional arguments passed to Playwright browserType.launch.
E.g. [args: ["--use-fake-ui-for-media-stream", "--use-fake-device-for-media-stream"]].
Can't be used with remote browser (ws_endpoint). The default value is [].

	:browser_launch_timeout (non_neg_integer/0) - The default value is 4000.

	:executable_path (String.t/0) - Path to a browser executable to run instead of the bundled one.
Use at your own risk.

	:headless (boolean/0) - The default value is true.

	:slow_mo (non_neg_integer/0) - The default value is 0.


The default value is [[id: :default_pool]].

	:ecto_sandbox_stop_owner_delay (non_neg_integer/0) - Delay in milliseconds before shutting down the Ecto sandbox owner after a
test ends. Use this to allow LiveViews and other processes in your app
time to stop using database connections before the sandbox owner is
terminated. The default value is 0.

	:executable_path (String.t/0) - Path to a browser executable to run instead of the bundled one.
Use at your own risk.

	:headless (boolean/0) - The default value is true.

	:js_logger (module | false) - false to disable, or a module that implements the PlaywrightEx.JsLogger behaviour. The default value is PhoenixTest.Playwright.JsLogger.

	:screenshot (boolean/0 | Keyword.t/0) - Either a boolean or a keyword list:
	:full_page (boolean/0) - The default value is true.

	:omit_background (boolean/0) - The default value is false.


The default value is false.

	:screenshot_dir (String.t/0) - Directory where screenshot/3 saves files. See also :snapshot_dir for assert_screenshot/3 reference images. The default value is "screenshots".

	:selector_engines - Define custom Playwright selector engines. The default value is [].

	:slow_mo (non_neg_integer/0) - The default value is 0.

	:snapshot_dir (String.t/0) - Directory where assert_screenshot/3 reads and writes baseline images.
Diff images on mismatch are written to <snapshot_dir>/__diff__/. Add that subdirectory to .gitignore:
**/__diff__/
The default value is "test/snapshots".

	:timeout (non_neg_integer/0) - The default value is 2000.

	:trace (boolean/0 | :open) - The default value is false.

	:trace_dir (String.t/0) - The default value is "traces".

	:ws_endpoint (String.t/0) - WebSocket endpoint URL for connecting to a remote Playwright server.
If provided, uses WebSocket transport instead of spawning a local Node.js process.
Example: "ws://localhost:3000/ws"
The remote server provides a single pre-launched browser, so browser_pool should
be set to false (pooling has no effect with a remote server).
This is useful for:
	Alpine Linux containers (glibc issues with local Playwright driver)
	Containerized CI environments with a separate Playwright server
	Connecting to remote/shared Playwright instances



Options that be overridden per test module via the use PhoenixTest.Playwright.Case opts:
	:browser_pool
	:browser
	:browser_launch_opts
	:browser_launch_timeout
	:executable_path
	:headless
	:slow_mo

Options that be overridden per test via ExUnit @tag:
	:accept_dialogs
	:ecto_sandbox_stop_owner_delay
	:screenshot
	:trace
	:browser_context_opts
	:browser_page_opts


      




  

    
PhoenixTest.Playwright.JsLogger 
    



      
Default javascript logger.

      




  

    
mix phoenix_test_playwright.test.websocket 
    



      
Runs the test suite against a Playwright server running in a Docker container,
connected via websocket transport.
This is useful for testing the websocket transport path without requiring a
local Playwright installation.
Usage
mix phoenix_test_playwright.test.websocket
mix phoenix_test_playwright.test.websocket --warnings-as-errors
mix phoenix_test_playwright.test.websocket test/specific_test.exs
All arguments are passed through to mix test.

